T1707205-v1 BS PI Matrix Calc - Test Hang Version 6/29/2017

Current version of the calculation assumes the BS Test Hang assembly frame arrangement with L (longitudinal) = -X (global) = y (this calculation) and T (transverse) = -Y = -x.

- **Graphics/geometry stuff**

```
vals = {
   R -> 715 / 1000, (* PI radius *)
   Sradius -> Sqrt[123.346^2 + 459.968^2] / 1000, (* 3D CAD *)
   Gradius -> 602.6 / 1000, (* 3D CAD *)
   Lradius -> Sqrt[131.764^2 + 579.648^2] / 1000, (* 3D CAD *)
   l -> 200 / 1000, (* For graphics *)
   origin -> 0 °
  };
```

```
Lradius /. vals
```

```
0.594435
```

- **Angular locations of the various items**

```
Sangles = {270 °, (270 + 120) °, (270 + 240) °}
```

```
{270 °, 390 °, 510 °}
```

```
Gangles = {315 °, (315 + 120) °, (315 + 240) °}
```

```
{315 °, 435 °, 555 °}
```

```
Langles = {330 °, (330 + 120) °, (330 + 240) °}
```

```
{330 °, 450 °, 570 °}
```

- **Positions of the various items as vectors**

```
Spositions = Sradius * Table[RotationMatrix[Sangles[[i]]].{1, 0}, {i, 1, 3}]
```

$$\left\{\{0, -\text{Sradius}\}, \left\{\frac{\sqrt{3}\ \text{Sradius}}{2}, \frac{\text{Sradius}}{2}\right\}, \left\{-\frac{\sqrt{3}\ \text{Sradius}}{2}, \frac{\text{Sradius}}{2}\right\}\right\}$$

```
Gpositions =  Gradius * Table[RotationMatrix[Gangles[[i]]].{1, 0}, {i, 1, 3}]
```

$$\left\{\left\{\frac{\text{Gradius}}{\sqrt{2}}, -\frac{\text{Gradius}}{\sqrt{2}}\right\}, \left\{\frac{\left(-1+\sqrt{3}\right)\text{Gradius}}{2\sqrt{2}}, \frac{\left(1+\sqrt{3}\right)\text{Gradius}}{2\sqrt{2}}\right\}, \left\{-\frac{\left(1+\sqrt{3}\right)\text{Gradius}}{2\sqrt{2}}, -\frac{\left(-1+\sqrt{3}\right)\text{Gradius}}{2\sqrt{2}}\right\}\right\}$$

```
Lpositions =  Lradius * Table[RotationMatrix[Langles[[i]]].{1, 0}, {i, 1, 3}]
```

$$\left\{\left\{\frac{\sqrt{3}\ \text{Lradius}}{2}, -\frac{\text{Lradius}}{2}\right\}, \{0, \text{Lradius}\}, \left\{-\frac{\sqrt{3}\ \text{Lradius}}{2}, -\frac{\text{Lradius}}{2}\right\}\right\}$$

- **Orientations of the various items as unit vectors**

```
Svectors = Table[RotationMatrix[Sangles[[i]]].{0, 1}, {i, 1, 3}]
```

$$\left\{\{1, 0\}, \left\{-\frac{1}{2}, \frac{\sqrt{3}}{2}\right\}, \left\{-\frac{1}{2}, -\frac{\sqrt{3}}{2}\right\}\right\}$$

```
Gvectors = Table[RotationMatrix[Gangles[[i]]].{0, 1}, {i, 1, 3}]
```

$$\left\{\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{-\frac{1+\sqrt{3}}{2\sqrt{2}}, \frac{-1+\sqrt{3}}{2\sqrt{2}}\right\}, \left\{\frac{-1+\sqrt{3}}{2\sqrt{2}}, -\frac{1+\sqrt{3}}{2\sqrt{2}}\right\}\right\}$$

```
Lvectors = Table[RotationMatrix[Langles[[i]]].{0, 1}, {i, 1, 3}]
```

$$\left\{\left\{\frac{1}{2}, \frac{\sqrt{3}}{2}\right\}, \{-1, 0\}, \left\{\frac{1}{2}, -\frac{\sqrt{3}}{2}\right\}\right\}$$

- **Diagram**

```
Graphics[{
   Circle[{0, 0}, R],
   Red,
   Sequence[Table[Arrow[{Spositions[[i]] - Svectors[[i]] * l / 2,
        Spositions[[i]] + Svectors[[i]] * l / 2}], {i, 1, 3}
    ]],
   Green,
   Sequence[Table[Arrow[{Gpositions[[i]] - Gvectors[[i]] * l / 2,
        Gpositions[[i]] + Gvectors[[i]] * l / 2}], {i, 1, 3}
    ]],
   Blue,
   Sequence[Table[Arrow[{Lpositions[[i]] - Lvectors[[i]] * l / 2,
        Lpositions[[i]] + Lvectors[[i]] * l / 2}], {i, 1, 3}
    ]]
  }] /. vals
```



```
(LVDTfromLTY = {
    Table[Lvectors[[i]].{1, 0}, {i, 1, 3}],
    Table[Lvectors[[i]].{0, 1}, {i, 1, 3}],
    Table[Lvectors[[i]].Lvectors[[i]] * Lradius, {i, 1, 3}]
   }) // TableForm
```

$\frac{1}{2}$	-1	$\frac{1}{2}$
$\frac{\sqrt{3}}{2}$	0	$-\frac{\sqrt{3}}{2}$
Lradius	Lradius	Lradius

```
LVDTfromLTY /. vals // N // TableForm
```

0.5	-1.	0.5
0.866025	0.	-0.866025
0.594435	0.594435	0.594435

```
(LTYfromLVDT = Inverse[LVDTfromLTY]) // TableForm
```

$\frac{1}{3}$	$\frac{1}{\sqrt{3}}$	$\frac{1}{3\,\text{Lradius}}$
$-\frac{2}{3}$	0	$\frac{1}{3\,\text{Lradius}}$
$\frac{1}{3}$	$-\frac{1}{\sqrt{3}}$	$\frac{1}{3\,\text{Lradius}}$

```
Transpose[LTYfromLVDT] /. vals // N // TableForm
```

0.333333	-0.666667	0.333333
0.57735	0.	-0.57735
0.560756	0.560756	0.560756

```
LTYfromLVDT /. vals // N // TableForm
```

0.333333	0.57735	0.560756
-0.666667	0.	0.560756
0.333333	-0.57735	0.560756